

User Experience SAS/GRAPH® Charts & Plots Using MXG®

Neal Musitano Jr.
Department of Veterans Affairs
Philadelphia Benefits Delivery Center
Philadelphia, Pennsylvania

ABSTRACT

IBM mainframe MVS computer performance data is collected using IBM System Management Facilities (SMF) and Resource Measurement Facility (RMF) data. The raw SMF/RMF data is processed with MXG® and SAS® software into an easily usable SAS® format.

This paper utilizes that performance data. The MXG® Performance Database, i.e., daily, weekly or monthly contains millions of performance variable values all in SAS® readable format.

The intent of this paper is not to explain MXG® software, which is SAS® source code, but to display some user examples of SAS/GRAPH® computer performance graphs utilizing the MXG® Computer Performance Database as input.

The SAS/GRAPH® Charts and Plots displayed were produced on the mainframe using the Computer Graphics Metafile driver “CGMMWWA” for Microsoft Word or PowerPoint. The GOPTION of GACCESS=GSASFILE is used to write the graph to a file on the mainframe. The graph is then downloaded to a PC in binary format, and finally inserted as a picture graphics into Microsoft® PowerPoint or Word documents.

INTRODUCTION

The IBM SMF/RMF performance data is processed by the SAS/MXG software into daily, weekly and monthly MXG® Computer Performance Databases.

These databases contain numerous SAS® datasets. Each SAS® dataset contains performance variables in a SAS® readable format. The advantage of using MXG® software is that it converts raw performance data from IBM and third party computer vendors into SAS® readable format for processing by SAS® jobs.

MXG® provides numerous examples of SAS® plots and charts, from utilizing SAS/BASE to using SAS/GRAPH®. However, each site is somewhat unique

in what charts are of most interest to management and how management wants the data presented.


The SAS/GRAPH® charts and plots produced for this presentation, utilize some of the basic MXG® DOCGRAF and GRAFXXXX examples as starting source code, but are then edited, tailored and modified to produce charts of interest to our site and kept to a minimum of code for this presentation.

BRIEF OVERVIEW

The following chart in Figure#1 displays the process of inputting the performance data into the MXG® build process to build the daily MXG® Performance Database or PDB.

The SAS/Graph® charts and plots that are displayed used the MXG® PDB as input.

Figure #1.



The objective was to utilize the MXG® performance databases to monitor system performance by producing SAS/GRAPH® charts and plots, that can easily be viewed by technicians or management.

The charts and plots were created on the mainframe, downloaded in binary format to a PC and inserted as a graphics picture, i.e. CGM file type into Microsoft® PowerPoint for e-mail distribution and presentations.

PREPARATION

The SAS/GRAPH® Charts and Plots displayed were produced on the mainframe using the Computer Graphics Metafile driver CGMMWWA for Microsoft Word. The graph is then downloaded to a PC in binary format.

The “userid.cgm” that the chart is written to on the mainframe is a physical sequential file, with a logical record length of 80 and a large or half track blocking. This is a small file of only two (2) tracks on a 3380 DASD.

Once the file was downloaded to a personal computer in binary format it had to be renamed to insure it had a file type of “CGM”. Then it could be inserted as a picture graphics into Microsoft PowerPoint (with the retain background option to keep the color backgrounds) or Microsoft Word documents.

SAS® CODE & CHARTS.

The RMFINTRV dataset contains hundreds of performance variables in one dataset. This makes processing performance data easier, since this dataset combines information from other PDB detailed datasets.

The charts can be created using as input a daily, weekly or monthly MXG® Performance database or an MXG® Trend database.

The SAS® code can be changed to select different date and time filters. With some additional coding, any performance variable can be charted for monitoring or to investigate all or any performance items of interest.

The following charts and plots display the system-id, the date or month and year. This variable data is taken from the input data in the MXG Performance Database. Since, the performance data is sorted by system-id and timestamp, it is possible to display the system-id and dates in the charts by using the “#BYVAL” feature.

There are over 80,000 MXG® software performance variables tracked. The charts in this paper display only a few of the basics, i.e., CPU, DASD, tape and TSO response time.

Figure #2 GCHART Procedure

```
//PDB DD DSN=MXG.PDB.MONTHLY(0),DISP=SHR
//SYSIN DD *
1 /* MEMBER=CPUCHART CHART CPU BY HOUR INPUT=RMFINTRV */
2 OPTIONS NOBYLINE;
3
4 DATA CPU24HR;
5 SET PDB.RMFINTRV;
6 IF SYSTEM NE 'SY1' THEN DELETE; /* ONLY CHART SYSTEM1'
7 IF DATE NE '10FEB1998'D THEN DELETE; /* LOOK AT FEB 10 '
8 RUN;
```

NOTE: The data set WORK.CPU24HR has 96 observations and 403 variables.
NOTE: The DATA statement used 0.62 CPU seconds and 3441K.

```
9
10 FILENAME GSASFILE 'userid.cgm';
11 GOPTIONS GUNIT=PCT RESET=GLOBAL DEVICE=CGMMWWA
12 FTEXT=ZAPF HTEXT=1.6 HANDSHAKE=NONE CBACK=VPAG
13 GACCESS=GSASFILE GSFLN=80 GSFMODE=REPLACE;
14
15 PROC GCHART DATA=CPU24HR; BY SYSTEM DATE;
16 FORMAT DATE WEEKDATE29.;
17 FORMAT PCTCPUBY 3.0;
18
19 TITLE1 HEIGHT=3.0 '#BYVAL1' 'COMPUTER SYSTEM';
20 TITLE2 HEIGHT=2.5 'CPU UTILIZATION FOR 24 HOURS';
21 TITLE3 H=2.5 C=RED '#BYVAL2';
22
23 AXIS1 WIDTH=1 C=BLACK LABEL=(HEIGHT=2.3 C=RED
24 ROTATE=90 ANGLE=-90 'PERCENT*CPU*UTILIZATION')
25 MAJOR=(HEIGHT=2 COLOR=GREEN)
26 MINOR=(H=1 N=4 COLOR=RED)
27 ORDER=(0 TO 100 BY 10)
28 VALUE=(HEIGHT=1.7 COLOR=BLUE);
29
30 AXIS2 WIDTH=3 COLOR=GOLD /*HORIZONTAL HOUR AXIS*/
31 LABEL=(HEIGHT=2.2 COLOR=BLUE)
32 VALUE=(HEIGHT=1.8 COLOR=BLACK);
33
34 PATTERN1 VALUE=SOLID COLOR=MAGENTA;
35
36 FOOTNOTE1 HEIGHT=1.8
37 J=L 'SOURCE: MXG PERFORMANCE DATABASE'
38 J=R 'SAS/GRAPH CHART';
39
40 FOOTNOTE2 HEIGHT=1.1 COLOR=RED
41 J=R 'CPUCHART';
42
43 VBAR HOUR / TYPE=MEAN SUMVAR=PCTCPUBY DISCRETE
44 RAXIS=AXIS1 MAXIS=AXIS2 MEAN REF=90;
45 RUN;
```

NOTE: 78 RECORDS WRITTEN TO GSASFILE.

NOTE: The above message was for the following by-group:
SYSTEM>ID=SY1 DATE*OF*STARTIME=Tuesday, February 10, 1998

Figure #3 is a SAS/GRAPH® vertical bar chart of average hourly CPU utilization.


The chart was created using the SAS/GRAPH® GCHART procedure and code in Figure #2.

Input to the program was the February 1998 monthly MXG® Performance Database, and a date filter was used to select the day desired for the chart.

The MXG® PDB.RMFINTRV dataset was used as input as can be seen in the set statement on line #5.

The RMFINTRV dataset contains hundreds of time-stamped performance variables. The value for the date variable is taken from the input data and is displayed in the chart title by using the #BYVAL2 variable value.

Figure #3 Hourly CPU Usage



SAS/GRAPH® Vertical Bar Chart

The SAS/GRAPH® Plot in Figure #5 was created with the SAS/GRAPH® GPLOT procedure.

This plot was created using the code in Figure#4.

This is a detailed plot of CPU utilization throughout the day for all four processors on system ‘SY1’.

This example uses a system and date filter to chart only system ‘SY1’ for one day. Again the monthly MXG Performance Database for February 1998 was used as input.

The date is taken from the input data. It is displayed in SAS Weekdate29 format as the third line of the chart title.

The detailed CPU usage data is from the MXG® PDB.TYPE70 dataset.


Figure #4 GPLOT Procedure

```

1  /* MEMBER=CPU2DAY DETAILED CPU USAGE BY PROCESSOR */
2  OPTION NOBYLINE;
3
4  DATA TYPE70;
5  SET PDB.TYPE70;
6  IF SYSTEM NE 'SY1' THEN DELETE; /*LOOK AT SY1*/
7  DATE = DATEPART(STARTTIME);
8  IF DATE NE '10FEB1998'D THEN DELETE;
9  TIME = TIMEPART(STARTTIME);
10 RUN;
11
12 FILENAME GSASFILE 'userid.cgm';
13 GOPTIONS GUNIT=PCT RESET=GLOBAL DEVICE=CGMMWWA
14 HANDSHAKE=NONE HTEXT=1.5 FTEXT=ZAPF CBACK=LIPK
15 GACCESS=GSASFILE GSFLN=80 GSFMODE=REPLACE;
16
17 PROC GPLOT DATA=TYPE70; BY SYSTEM DATE;
18 FORMAT DATE WEEKDATE29.;
19 FORMAT TIME TIME5.0;
20 LABEL TIME='TIME*OF*DAY';
21
22 TITLE1 H=3.0 '#BYVAL1' 'COMPUTER SYSTEM';
23 TITLE2 H=2.5 'DETAILED CPU ACTIVITY BY PROCESSOR';
24 TITLE3 H=2.5 C=BLUE '#BYVAL2';
25
26 AXIS1 WIDTH=2 COLOR=BLUE /*VERTICAL CPU AXIS*/
27 LABEL=(H=2.4 COLOR=RED ROTATE=90 ANGLE=-90
28 'PERCENT*CPU')
29 MAJOR=(HEIGHT=2 COLOR=GREEN)
30 MINOR=NONE
31 ORDER=(0 TO 100 BY 10)
32 VALUE=(H=1.7 COLOR=BLUE);
33
34 AXIS2 /*HORIZONTAL TIME AXIS*/
35 LABEL=(HEIGHT=2.2 COLOR=GREEN)
36 MAJOR=(HEIGHT=2 COLOR=RED)
37 MINOR=(H=1 N=5 C=BIB)
38 ORDER=('00:00'T TO '24:00'T BY '06:00'T)
39 VALUE=(H=2.5 COLOR=BLACK);
40
41 FOOTNOTE1 HEIGHT=1.8
42 J=L 'SOURCE: MXG PERFORMANCE DATABASE'
43 J=R 'SAS/GRAPH PLOT';
44
45 FOOTNOTE2 HEIGHT=1 COLOR=RED J=R 'CPU2DAY';
46
47 SYMBOL1 INTERPOL=JOIN C=BLACK L=1 W=1 VALUE=NONE;
48 SYMBOL2 I=JOIN C=RED L=2 W=1 V=SQUARE H=2;
49 SYMBOL3 I=JOIN C=BLUE L=3 W=1 V=CIRCLE H=2;
50 SYMBOL4 I=JOIN C=GREEN L=4 W=1 V=TRIANGLE H=2;
51
52 LEGEND1 VALUE=(HEIGHT=2.2) ACROSS=2
53 CFRAME=WH CBORDER=BLUE LABEL=NONE;
54
55 PLOT PCTCPBY1*TIME PCTCPBY2*TIME
56 PCTCPBY3*TIME PCTCPBY4*TIME / OVERLAY
57 HAXIS=AXIS2 VAXIS=AXIS1 CFRAME=WH
58 VREF=90 CVREF=GREEN LEGEND=LEGEND1;
59 RUN;

```

Figure #5 Detailed CPU Use



SAS/GRAPH® Plot

Figure #6 GCHART Procedure

```

1  /** MEMBER=CPUDATE CHART DAILY CPU FOR THE MONTH **/
2  OPTION NOBYLINE;
3  PROC FORMAT; VALUE MONFM
4  1='JANUARY' 2='FEBRUARY' 3='MARCH' 4='APRIL'
5  5='MAY' 6='JUNE' 7='JULY' 8='AUGUST'
6  9='SEPTEMBER' 10='OCTOBER' 11='NOVEMBER' 12='DECEMBER';

8  DATA RMFINTRV;
9  FORMAT MONTH MONFM.;
10 SET PDB.RMFINTRV;
11 IF SYSTEM NE 'SY1' THEN DELETE; /* ONLY SYSTEM SY1 */
12 IF TIMEPART(STARTTIME) LT '08:00:00'T THEN DELETE;
13 IF TIMEPART(STARTTIME) GE '16:00:00'T THEN DELETE;
14 MONTH = MONTH(DATE);
15 YEAR = YEAR(DATE);
16 DAY=DAY(DATE);
17 RUN;

NOTE: The data set WORK.RMFINTRV has 992 observations and 406 variables.
NOTE: The DATA statement used 0.84 CPU seconds and 3566K.

18
19 FILENAME GSASFILE 'userid.cgm';
20 GOPTIONS GUNIT=PCT RESET=GLOBAL DEVICE=CGMMWVA
21 NOPROMPT FTEXT=ZAPF HTEXT=1.8 HANDSHAKE=NONE CBACK=VPAB
22 GACCESS=GSASFILE GSFLN=80 GSFMODE=REPLACE;
23
24 PROC GCHART DATA=RMFINTRV; BY SYSTEM MONTH YEAR;
25 FORMAT DAY 3.0;
26 FORMAT PCTCUBY 4.0;
27 LABEL DAY='DAY*OF*MONTH';
28
29 TITLE1 HEIGHT=3.0 '#BYVAL1' 'COMPUTER SYSTEM';
30 TITLE2 HEIGHT=2.8 'CPU UTILIZATION 08:00 - 16:00';
31 TITLE3 C=RED HEIGHT=2.8 '#BYVAL2 ' ' #BYVAL3';
32
33 AXIS1 /*CPU USAGE AXIS*/
34 LABEL=(HEIGHT=2.2 COLOR=BLACK)
35 MAJOR=(HEIGHT=2.0 COLOR=GREEN)
36 MINOR=(HEIGHT=1.0 COLOR=BLACK N=4)
37 ORDER=(0 TO 100 BY 10)
38 VALUE=(H=2.0 COLOR=RED);
39
40 AXIS2 WIDTH=6 COLOR=GOLD /*DAY AXIS*/
41 LABEL=(HEIGHT=2.0 C=RED ROTATE=90 ANGLE=-90)
42 VALUE=(COLOR=BLACK H=1.5);
43
44
45 FOOTNOTE1 H=1.8 J=L 'SOURCE: MXG PERFORMANCE DATABASE'
46 J=R 'SAS/GRAPH CHART';
47 FOOTNOTE2 H=1.0 J=R 'CPUDATE';
48
49 PATTERN1 COLOR=BLUE V=SOLID;
50
51 HBAR DAY / SUMVAR=PCTCUBY TYPE=MEAN DISCRETE
52 RAXIS=AXIS1 MAXIS=AXIS2 CFRAME=WH MEAN;
53 RUN;

```

The SAS® code in Figure#6 is used to create the SAS/GRAPH® Chart of average CPU utilization for the day shift -- Figure #7.


The chart is for system 'SY1'. The month and year are displayed on the chart and this data is taken as input from the PDB.RMFINTRV

dataset in the monthly MXG® Performance Database.

The SAS Proc Format was used to assign the month a more user friendly value instead of using the numerical values of 1 through 12.

The chart provides a quick view that CPU usage is generally consistent throughout the month, with the weekends showing much lighter CPU usage.

Figure #7 Monthly CPU Usage



SAS/GRAPH® Horizontal Bar Chart

Figure #8 Gplot Procedure

```

1  /** MEMBER=T74DAY  PLOT DASH RESPONSE **/
2  OPTIONS NOBYLINE;
3
4  DATA INFO74;
5  SET PDB.TYPE74;
6  IF SYSTEM NE 'SY1' THEN DELETE;
7  IF SUBSTR(VOLSER,1,5) NE 'SPOOL' THEN DELETE;
8  TIME=TIMEPART(STARTTIME);
9  DATE=DATEPART(STARTTIME);
10 RUN;

NOTE: The data set WORK.INFO74 has 384 observations and 103 variables.
NOTE: The DATA statement used 0.51 CPU seconds and 3211K.

11
12  FILENAME GSASFILE 'userid.cgm';
13  GOPTIONS GUNIT=PCT RESET=GLOBAL DEVICE=CGMMWWA
14  HANDSHAKE=NONE FTEXT=ZAPF HTEXT=1.5 CBACK=VPAV
15  GACCESS=GSASFILE GSFLN=80 GSFMODE=REPLACE;
16
17  PROC Gplot DATA=INFO74; BY SYSTEM DATE;
18  FORMAT DATE WEEKDATE29.;
19  FORMAT TIME TIME5.0;
20  TITLE1 HEIGHT=3.0 '#BYVAL1' 'COMPUTER SYSTEM';
21  TITLE2 HEIGHT=2.5 'DASH I/O RESPONSE TIME';
22  TITLE3 HEIGHT=2.5 '#BYVAL2';
23
24  SYMBOL1 INTERPOL=JOIN C=GREEN L=1 W=2 V=NONE;
25  SYMBOL2 INTERPOL=JOIN C=RED L=2 W=2 V=SQUARE H=3;
26  SYMBOL3 INTERPOL=JOIN C=BLUE L=3 W=2 V=CIRCLE H=3;
27  SYMBOL4 INTERPOL=JOIN C=BLACK L=4 W=2 V=STAR H=3;
28
29  AXIS1 WIDTH=6 COLOR=MAGENTA /*DASH RESPONSE*/
30  LABEL=(HEIGHT=1.7 C=BLACK ROTATE=90 ANGLE=-90)
31  MAJOR=(HEIGHT=2 COLOR=GREEN) MINOR=NONE
32  VALUE=(HEIGHT=1.7 COLOR=BLUE);
33
34  AXIS2 /*HORIZONTAL TIME AXIS*/
35  LABEL=(HEIGHT=2.2 COLOR=GREEN)
36  MAJOR=(H=2 C=GREEN) MINOR=(H=1 N=5 COLOR=BLUE)
37  ORDER=(00:00'T TO 24:00'T BY 06:00'T)
38  VALUE=(HEIGHT=2.5 COLOR=RED);
39
40  FOOTNOTE1 H=1.8 J=L
41  'SOURCE: MXG PERFORMANCE DATABASE'
42  J=R 'SAS/GRAPH PLOT';
43  FOOTNOTE2 H=1.0 J=R COLOR=RED 'T74DAY';
44
45  LEGEND1
46  LABEL=(H=2.0 C=MAGENTA POSITION=(BOTTOM CENTER))
47  ACROSS=2 VALUE=(H=2.0 COLOR=GOLD)
48  CBORDER=BLUE CFRAME=WH CSHADOW=GOLD;
49
50  PLOT AVGRSPMS*TIME=VOLSER / LEGEND=LEGEND1
51  VAXIS=AXIS1 HAXIS=AXIS2 CFRAME=PAY;
52  RUN;

NOTE: 834 RECORDS WRITTEN TO GSASFILE.

```

Figure #9 is a SAS/GRAPH® Plot of DASH response time for the SPOOL DASH on system 'SY1'.

The plot was created using the code in Figure #8.


A filter was used to select only the SPOOL DASH

volumes for the chart - reference line #7.

The input to the program was the PDB.TYPE74 dataset in the daily MXG® Performance Database. The daily MXG® PDB is built just after the RMF/SMF data is dumped at midnight.

DASH response can vary when an I/O problem develops. Thus, the AXIS1 statement in the example did not include a specific order of data values to prevent "clipping" of data out of range and let SAS® pick the axis size.

Figure #9 DASH RESPONSE



CONCLUSION

Utilizing the MXG® daily, weekly or monthly performance databases as input to the SAS/GRAPH® programs provides high quality graphical displays of computer performance data that are easily interpreted by tactical performance technicians for investigating “bottlenecks” or by management to review overall system performance.

Downloading and inserting the graphs into Microsoft® PowerPoint and Word for e-mail and presentations has increased the use of MXG® and SAS/GRAPH®. This is due to the detail of information in the graphs and the use of the graphs in more computer performance reviews and management reports.

REFERENCES

IBM OS/390 - MVS Initialization and Tuning Reference - SC28-1752-02.

IBM OS/390 - MVS System Management Facilities (SMF) - GC28-1783-02

IBM OS/390 - Resource Measurement Facility (RMF) User's Guide - SC28-1949-01.

Merrill's Expanded Guide to Computer Performance Evaluation Using the SAS System.

Merrill's Expanded Guide Supplement.

Technical Newsletters for Users of MXG®.

NESUG Proceeding 1992. Using SAS/Graph® Output in Non-SAS Programs, by Michael S. Zdeb.

NESUG Proceeding 1994. Effective Poster Presentations for SAS® User Group Conferences: Helen-Jean Talbott

SAS® User's Guide: Basics.

SAS/GRAPH® User's Guide.

SAS/Graph® Software: Using Graphics Devices in the MVS Environment.

SAS/GRAPH® Software: Reference, Version 6.

Neal Musitano Jr.
Department of Veterans Affairs
Benefits Delivery Center-284/31
P. O. BOX 7545
Philadelphia, PA. 19101-7545

Phone 215-842-2000 ext. 1-4102
FAX 215-381-3456
EMAIL TSDNMUSI@VBA.VA.GOV

Author Contact